

Crossword Generator and Solver

Gustaf Ahdritz (gwa2107) and Lucie le Blanc (ll3163)

23 December 2020

1 Introduction and Overview

This project creates crossword grids and solves them using a user-specified dictionary. There are no clues or pre-filled letters involved: the solver produces all valid solutions for the crossword based on its dimensions and layout.


2 Crossword generation

The crossword grid generation loosely follows the New York Times' crossword construction constraints¹:

1. Crosswords must have black square symmetry, which typically comes in the form of 180-degree rotational symmetry;
2. Crosswords must have all-over interlock;
3. Crosswords must not have unchecked squares (i.e., all letters must be found in both Across and Down answers);
4. All answers must be at least 3 letters long;
5. Black squares should be used in moderation.

Constraints 1 and 3 are fully implemented: our crosswords have 180-degree rotational symmetry, and all white squares are part of both horizontal and vertical entries. We implemented constraints 4 and 5 by allowing our crossword generator to take in a user-supplied minimum word length and target black square density. The rotational symmetry constraint made it very difficult to generate crosswords with odd dimensions that didn't violate the minimum word length, so we set our crossword generator to only generate grids with even dimensions. Our strategy also means that it is possible to generate grids with disconnected regions, which breaks constraint 2. See below for one of the example grids we generated, using a size of 10 and a black square density of 0.2.

¹<https://www.nytimes.com/puzzles/submissions/crossword>



3 Solving strategy

The solving strategy for the crossword is straightforward. Using a user-supplied dictionary, solution candidates are generated by filling each row of the grid with possible word combinations. The validity of the candidate is checked column-wise. Valid solutions are counted and printed.

Internally, grids are represented as two-dimensional lists of `Square`. Filling the crossword involves using list comprehensions to generate all possible solutions for each word, row, and grid. Checking the crossword is done by transposing each solution candidate and checking each column-turned-row to make sure it only contains sequences of letters found in the dictionary.

4 Dictionary selection

In our initial attempts, the dictionary was stored as a `Map Int (Set String)`, where integer keys representing word length mapped to sets of words with that number of letters. This was later changed to `Map Int (Set [Square])`, for reasons discussed later in this report. The `Map/Set` combination was chosen for efficient lookup.

The standard dictionary, located in `/usr/share/dict/words`, contains 102774 words. 63357 of these contain no special characters or capital letters, making them usable in our crosswords. However, testing with a dictionary this large proved to be too difficult – every machine we tested on, including a CLIC machine called Den Haag that boasts 125G of memory, failed to get past even the first layer of candidates for a 4x4 grid before getting killed. The solutions it did generate were often quite interesting (see Figure 1).

[a b l y]	[a c i d]	[a c t s]
[b e a u]	[b o r e]	[b o r e]
[e a r l]	[e p i c]	[e v e n]
[d u k e]	[d e s k]	[d e e d]

Figure 1: 4x4 puzzles generated using standard dictionary.

We instead chose to test smaller dictionaries of varying sizes. We started with a dictionary of 10000 most common English words², and trimmed the list by taking only the first 1000, 500, 200 or 100 words for different tests.

In terms of crossword filling, this does not yield very interesting solutions. Words like “the” and “that” appear often, but would not be not easy to write creative clues for in a real puzzle. In the 1000-word case, filling a 4x4 grid generated 39 solutions, 37 of which were diagonally symmetrical (same words down as across). The other 2 of were each others’ transposes, and missed diagonal symmetry by only one letter.

[f a s t]	[s a l e]
[a u t o]	[a r e a]
[s t a y]	[l e s s]
[t o y s]	[e a s y]

Figure 2: 4x4 puzzles generated using 1000-common-word dictionary.

A more practical crossword generator would prefer words toward the middle of the frequency-ordered word list. Ideally, crossword words should be known to the average person but not necessarily used in the average sentence.

5 Initial Parallelization Attempts

Unless specified otherwise, all results in this section are from tests run using a dictionary of 500 randomly selected common words on the following 4x4 crossword:

$$\begin{bmatrix} @ & - & - & - \\ - & - & - & - \\ - & - & - & - \\ - & - & - & @ \end{bmatrix}$$

where underscores represent blank white spaces and ‘@’ represents a black square. While the solver can handle larger puzzles than this, increasing the size of the puzzle increases runtimes to the order of minutes rather than seconds, prohibiting analysis of event logs in Threadscope. In the following subsections, we will apply various parallelization techniques to the following (slightly squashed) code snippet:

²<https://github.com/first20hours/google-10000-english>

```

solve :: Map Int (Set String) -> [[Square]] -> [[[Square]]]
solve dict grid = do
  let cand = map transpose (fill dict grid)
      filter (check dict) cand

```

The third line yields a list of transposed candidate solutions. The fourth cross-references them against the provided dictionary to verify that the columns contain valid words.

All figures reported in this section represent an average of five trials on an otherwise lightly loaded system.

Run sequentially, the program finds all valid solutions in 11.07 seconds.

5.1 Naive parBuffer

Our first attempt at parallelization was simply to parallelize the checking, expecting each thread to generate and then check their respective candidates independently, as follows:

```

solve :: Map Int (Set String) -> [[Square]] -> [[[Square]]]
solve dict grid = do
  let cand = map transpose (fill dict grid)
      filter (check dict) cand `using` parBuffer 100 rdeepseq

```

We use parBuffer here rather than parList in an attempt to mitigate the extreme memory requirements of loading the whole list of candidates at once and rdeepseq to force complete evaluation of each candidate in each thread. In practice, this implementation failed to parallelize at all. Zero threads are created,




Figure 3: The naive parBuffer implementation, -N2

and the overhead of adding a second core (on which some of the garbage collection seems to be running) slows the runtime to 14.44s. The ThreadScope graph is included in Figure 3. The culprit here appears to be filter. Unlike map, whose weak head normal form representation would consist of a concatenation of one thunk for applying the check to the first element of the list and then another for completing the map operation on the remainder of the list, filter must evaluate each check on each element completely in order to determine the shape of

the list. This evaluation is performed sequentially before the `parBuffer` is even properly applied, completely negating the benefits of having multiple cores.

5.2 Parallelizing candidate creation

We can sidestep this problem by shifting the parallelization directly onto the code that generates candidate solutions:

```
solve :: Map Int (Set String) -> [[Square]] -> [[[Square]]]
solve dict grid = do
  let l = fill dict grid
      cand = map transpose l `using` parBuffer 100 rdeepseq
      filter (check dict) cand
```

The event log tells us that this change was successful: The workload here appears




Figure 4: `parBuffer` applied to candidates, -N2

extremely well-balanced. However, since about 67% of the 11694774 sparks created either fizzle or are garbage collected, sequential garbage collection for excess sparks short enough not to be visible at this scale in Threadscope drives the runtime all the way up to 24.09s, most of which comes in the form of additional garbage collection: we have achieved a “speedup” of 0.46x. This indicates that the parallelization is far too granular, making it inefficient: each spark is performing much less work per candidate grid than its computational upkeep.

5.3 Parallelized candidate creation with chunking

Instead of parallelizing the output of `map` directly, we can group the candidates into chunks of more reasonable sizes and then hand off each one to be computed in parallel:

```
solve :: Map Int (Set String) -> [[Square]] -> [[[Square]]]
solve dict grid = do
  let l = fill dict grid
      m = map transpose l
```

```

let chunk n = Data.List.Split.chunksOf n
let cand = chunk 10000 m `using` parBuffer 100 rdeepseq
filter (check dict) $ concat cand

```




Figure 5: parBuffer applied to candidates w/ chunking, -N4

On this test, we see a 100% conversion rate: of all of the 19617 sparks created, none fail. The program spends much less time in the MUT phase of computation as a result, but the costs of garbage collection gives us our slowest runtime so far: 31.6s. Perturbing the chunk size to any of {10, 50, 100, 500, 1000, 10000, 50000, 100000} or changing the number of threads passed to parBuffer makes matters worse.

5.4 Parallelizing candidate creation & filtering

Instead of filtering candidates in a separate second step, we can check them at the same time:

```

solve :: Map Int (Set String) -> [[Square]] -> [[[Square]]]
solve dict grid = do
  let l = fill dict grid
      m = map transpose l
      f = filter (check dict)
      chunk n = Data.List.Split.chunksOf n
      chunks = chunk 20000 m
      concat $ map f chunks `using` parBuffer 100 rdeepseq

```

Note that we have heuristically increased the chunk size to 20000. This is a substantial improvement over the previous attempt; run on 4 cores. The

conversion rate for the 981 sparks created is 100%. The MUT phase of the program runs in just 3.57s. Unfortunately, the greater the number of cores, the greater the time spent on garbage collection. The total runtime on 4 cores is 16.36s—faster than in the previous section, but still not close to the 11.07s achieved sequentially.




Figure 6: parBuffer applied to both candidate generation and filtering, -N4

Increasing or decreasing the core count, curiously enough, both have the effect of increasing the absolute time spent on garbage collection. The problem is only exacerbated on more complex tasks. When the size of the dictionary is increased to 1000, the parallel -N4 solver spends about 85 seconds in MUT and then another 285 in garbage collection, for a total of 370. The sequential solver using the same dictionary spends 242 seconds in MUT and then 52 in garbage collection, for a smaller total of 294.³ Figure 6 shows that, like Marlow’s “kmeans_strat,” our program performs computations in staggered chunks separated by periods of protracted garbage collection. Though we can decrease the width of each individual stop by decreasing the chunk size, the total time spent on garbage collection remains fairly constant. Unlike Marlow, for whom the suspected culprit was expensive I/O operations, we make a point of not printing anything during execution. On the other hand, an inspection of Threadscope’s “Raw events” tab reveals a number of heap overflows in individual threads, causing needless interruptions. We conclude that the issue must be related to inefficient memory constructions in the crossword source.

6 Speedup Attempts

We made several changes to our program in an attempt to fix the very large garbage collection overhead.

³Due to their length, the 1000-word trials were only run once each

6.1 List Comprehension

One of our attempts to improve our solution involved changing how we generated solution candidates. Originally, we used a list comprehension and recursion when filling the grid.

```
fill_crossword dict (l:ls) =
    [ line : rest | line <- fill_line dict l,
      rest <- fill_crossword dict ls ]
```

At the suggestion of Prof. Edwards, we switched order of the list comprehension generators in the `fill crossword` and `fill line` functions. This has the effect of staggering memory-intensive recursive calls to `fill crossword` and reducing the likelihood of expensive heap overflows.

```
fill_crossword dict (l:ls) =
    [ line : rest | rest <- fill_crossword dict ls,
      line <- fill_line dict l ]
```

As will be shown in section 7, this change led to a speedup for both solvers, particularly the parallel one.

Another attempt involved generating candidates without a list comprehension at all. Instead, we used the `sequence` function, which takes lists as a parameter and returns their Cartesian product, to generate all possible solution candidate combinations.

```
fill_crossword dict grid = sequence result_grid
    where result_grid = Prelude.map (fill_line dict) grid
```

This modification led the program to consume even more memory and spend more time in garbage collection; we reverted it to the flipped list comprehension above.

6.2 Set and List Comparison

Another possible source of problems we identified was how words were being picked from the dictionary when building candidate grids. Instead of `Map Int (Set String)`, we changed the dictionary passed to the `fill` functions to be of type `Map Int [String]`, in order to avoid possible overhead from the `Set toList` function. This did not affect the speed or memory consumption, perhaps because the results of `Set toList` are already being efficiently cached.

6.3 Dictionary Word Storage

We attempted to speed up dictionary lookups by eliminating the conversion step from `String` to `Square` and back. This involved changing the dictionary type from `Map Int (Set String)` to `Map Int (Set [Square])`. This did not end up changing much, but it was one of the few cases in which the parallel version ran slightly faster than the sequential version.

7 Improved Parallelization Results

After implementing these improvements and dramatically shrinking the size of the chunks being passed to our sparks (from 10000 \rightarrow 50), we observe the results shown in Figure 7.




Figure 7: Parallelization strategy from 5.4 with improvements, -N4

Note the consistently higher activity on the cores and the absence of gaps as wide as those in Figure 6. Since the chunk size is so much smaller than the previous test, we see a dramatically increased number of sparks: about 392,000. As in earlier tests, however, all but 16 GC'd sparks and 24 fizzles successfully converted. For the first time, we are able to observe a speedup relative to the sequential version. The optimizations in Section 6 also benefited the latter: it now runs the task in 10.70s. However, the parallel version is even faster, finishing in 9.05s, 7 seconds faster than in the previous test. This buys us a final speedup of about $\times 1.18$. Granted, though the two allocate a similar number of bytes over their lifetimes—about 40 billion for the sequential version and 36 for the parallel—the productivity of the sequential version is much higher. It spends 97.1% of its time in MUT where the parallel version spends just 24.5%, leaving apparent room for future optimizations.

8 Discussion

While we were able to achieve high degrees of parallelization, the memory requirements of our program overwhelmed most of the accompanying performance benefits. However, the performance benefits we did see underneath the garbage collection overhead lead us to believe that further improvements are still possible.

A Source Code

A.1 Main.hs

```
1 import Data.Map
2 import Data.Set
3 import Data.List
4 import Data.List.Split
5 import Control.Parallel.Strategies
6 import System.Random
7 import System.Environment(getArgs, getProgName)
8 import System.Exit(die)
9 import DictUtil
10 import SquareDefs
11 import GenerateGrid
12
13 fill_and_check_s :: Map Int (Set [Square]) -> [[Square]] -> [[[Square]]]
14 fill_and_check_s dict grid = do
15     let grids = fill_crossword dict grid
16         candidates = Prelude.map transpose grids
17         Prelude.filter (check_crossword dict) candidates
18
19 fill_and_check_p :: Map Int (Set [Square]) -> [[Square]] -> [[[Square]]]
20 fill_and_check_p dict grid = do
21     let f = Prelude.filter (check_crossword dict)
22         grids = fill_crossword dict grid
23         m = Prelude.map transpose grids
24         c = Data.List.Split.chunksOf 50 m
25         p = parBuffer 100 rseq
26         sols = Prelude.map f c `using` p
27     concat sols
28
29 main :: IO ()
30 main = do args <- getArgs
31         case args of
32             [dimS, mS, threshS, dictPath, deterministic, parallel] -> do
33                 seed <- case deterministic of
34                     "y" -> return $ mkStdGen 15
35                     _   -> newStdGen
36                 let dim = read dimS :: Int
37                     m = read mS :: Int
38                     thresh = read threshS :: Float
39                     result = generateGrid dim m thresh seed
40                     peeled = maybe [[]] (\x -> x) $ result
41                     mapM_ (putStrLn) $ Prelude.map show peeled
42                     dictStream <- readFile dictPath
43                     let dict = build_dict dictStream
```

```

44         filled <- case parallel of
45             "y" -> return $ fill_and_check_p dict peeled
46             _   -> return $ fill_and_check_s dict peeled
47         mapM_ (printGrid) filled
48         let num = show (length filled)
49             putStrLn $ "found " ++ num ++ " ways to fill the given grid"
50     _ -> do pn <- getProgName
51             die $ "Usage: " ++ pn ++ " <dimension> <min_word_length> \
52                 \ <probability_threshold> \
53                 \ <dict_filepath> \
54                 \ <deterministic> \
55                 \ <parallel>"

```

A.2 SquareDefs.hs

```

1  module SquareDefs where
2
3  import Control.DeepSeq
4
5  data Square = Black | White Char
6
7  instance Show Square where
8      show Black = "@"
9      show (White c) = [c]
10
11 instance Eq Square where
12     (==) Black Black = True
13     (==) (White a) (White b) = (a == b)
14     (==) _ _ = False
15
16 instance NFData Square where
17     rnf s = s `seq` ()
18
19 instance Ord Square where
20     Black `compare` Black = EQ
21     (White a) `compare` (White b) = (a `compare` b)
22     Black `compare` _ = GT
23     _ `compare` Black = LT

```

A.3 DictUtil.hs

```

1  module DictUtil where
2
3  import Data.Set

```

```

4 import Data.Map
5 import Data.Char
6 import Data.List
7 import SquareDefs
8
9
10 blockedSq :: Char
11 blockedSq = 'X'
12
13 set_unwrap :: Maybe (Set [Square]) -> Set [Square]
14 set_unwrap (Just x) = x
15 set_unwrap Nothing = Data.Set.empty
16
17 getWordSet :: Map Int (Set [Square]) -> Int -> Set [Square]
18 getWordSet dict len = set_unwrap $ Data.Map.lookup len dict
19
20 pick_word :: Set [Square] -> [[Square]]
21 pick_word dict_slice = Data.Set.toList dict_slice
22
23 fill_word :: Map Int (Set [Square]) -> [Square] -> [[Square]]
24 fill_word _ [] = [[]]
25 fill_word dict single_word = pick_word candidateSet
26     where candidateSet = getWordSet dict $ length single_word
27
28 fill_line :: Map Int (Set [Square]) -> [Square] -> [[Square]]
29 fill_line _ [] = [[]]
30 fill_line dict xs = [ w ++ blocked ++ r | w <- fill_word dict firstWord,
31     r <- fill_line dict restOfLine ]
32     where (word, suffix) = break isBlocked xs
33           (blocked, restOfLine) = span isBlocked suffix
34           isBlocked Black = True
35           isBlocked (White _) = False
36
37
38 fill_crossword :: Map Int (Set [Square]) -> [[Square]] -> [[[Square]]]
39 fill_crossword _ [] = [[]]
40 fill_crossword dict (l:ls) = [ c : r | c <- fill_crossword dict ls,
41     r <- fill_line dict l ]
42
43 -- An alternative fill_crossword. In our tests, it did not perform as well
44 {- sequence result_grid
45     where result_grid = Prelude.map (fill_line dict) unfilled_grid -}
46
47 check_word :: Map Int (Set [Square]) -> [Square] -> Bool
48 check_word _ [] = True
49 check_word dict single_word = Data.Set.member single_word candidateSet
50     where candidateSet = getWordSet dict $ length single_word
51

```

```

52 check_line :: Map Int (Set [Square]) -> [Square] -> Bool
53 check_line _ [] = True
54 check_line dict xs = check_word dict firstWord && check_line dict restOfLine
55     where (firstWord, suffix) = break isBlocked xs
56           (_, restOfLine) = span isBlocked suffix
57           isBlocked Black = True
58           isBlocked (White _) = False
59
60 check_crossword :: Map Int (Set [Square]) -> [[Square]] -> Bool
61 check_crossword dict grid = Prelude.foldl (&&) True m
62     where m = Prelude.map (check_line dict) grid
63
64 check_crosswords :: Map Int (Set [Square]) -> [[[Square]]] -> Bool
65 check_crosswords dict grids = Prelude.foldl (&&) True m
66     where m = Prelude.map (check_crossword dict) grids
67
68 valid_dict_word :: String -> Bool
69 valid_dict_word w = Prelude.foldl (&&) True $ Prelude.map isAcceptable w
70     where isAcceptable c = isAscii c && isLower c
71
72 get_valid_words :: String -> [String]
73 get_valid_words streamIn = Prelude.filter valid_dict_word $ words streamIn
74
75 insert_word :: Map Int (Set [Square]) -> String -> Set [Square]
76 insert_word dict single_word = Data.Set.insert squarified_word len_set
77     where len_set = getWordSet dict $ length single_word
78           squarified_word = Prelude.map toSquare single_word
79           toSquare c = White c
80
81 update_set :: Map Int (Set [Square]) -> String -> Map Int (Set [Square])
82 update_set dict single_word = Data.Map.insert word_len updated_set dict
83     where word_len = length single_word
84           updated_set = insert_word dict single_word
85
86 build_dict_rec :: Map Int (Set [Square]) -> [String] -> Map Int (Set [Square])
87 build_dict_rec dict [] = dict
88 build_dict_rec dict (w:ws) = update_set (build_dict_rec dict ws) w
89
90 build_dict :: String -> Map Int (Set [Square])
91 build_dict streamIn = build_dict_rec Data.Map.empty valid_words
92     where valid_words = get_valid_words streamIn
93
94 fill_and_check :: Map Int (Set [Square]) -> [[Square]] -> [[[Square]]]
95 fill_and_check dict emptyGrid = Prelude.filter (check_crossword dict) candidates
96     where candidates = Prelude.map transpose f
97           f = fill_crossword dict emptyGrid

```

A.4 GenerateGrid.hs

```
1 module GenerateGrid where
2
3 import Data.List (sortBy)
4 import Data.Map.Strict (Map, (!))
5 import qualified Data.Map.Strict as Map
6 import Data.Set (Set)
7 import qualified Data.Set as Set
8 import System.Random
9 import SquareDefs
10
11 type Index = (Int, Int)
12 type Constraint = (Int, Int, Int, Int)
13 type Constraints = Map Index Constraint
14
15 probs :: RandomGen g => Int -> g -> [Float]
16 probs 0 _ = []
17 probs n s = r : probs (n-1) s'
18     where
19         (r, s') = randomR (0.0::Float, 1.0::Float) s
20
21 indices :: Int -> Int -> Set Index
22 indices h w = Set.fromList $ [(i,j) | i <- [0..(h - 1)], j <- [0..(w - 1)]]
23
24 initConstraints :: Int -> Int -> Constraints
25 initConstraints h w = Map.fromSet (\_ -> (0, w - 1, 0, h - 1)) (indices h w)
26
27 updateConst :: Index -> Index -> Constraint -> Constraint
28 updateConst (bi, bj) (i, j) (l,r,t,b) = (nl, nr, nt, nb)
29     where
30         (nl, nr) | not sameRow = (l, r)
31                 | left = (max l bj + 1, r)
32                 | right = (l, min r bj - 1)
33                 | otherwise = (i, j)
34         (nt, nb) | not sameCol = (t,b)
35                 | above = (max t bi + 1, b)
36                 | below = (t, min b bi - 1)
37                 | otherwise = (i, j)
38         sameRow = bi == i
39         (left, right) = (bj < j, bj > j)
40         sameCol = bj == j
41         (above, below) = (bi < i, bi > i)
42
43 adjustWithKeyList :: Ord k => (k -> a -> a) -> [k] -> Map k a -> Map k a
44 adjustWithKeyList _ [] m = m
45 adjustWithKeyList f (x:xs) m = Map.adjustWithKey f x $ adjustWithKeyList f xs m
```

```

46
47 addBS :: Index -> Constraints -> Constraints
48 addBS (i,j) con = updateRow $ updateCol con
49     where
50         updateCol c = adjustWithKeyList f col c
51         updateRow c = adjustWithKeyList f row c
52         f = updateConst (i,j)
53         col = filter (\(_,b) -> b == j) keyList
54         row = filter (\(a,_) -> a == i) keyList
55         keyList = Map.keys con
56
57 checkSquare :: Index -> Constraints -> Int -> Bool
58 checkSquare (i,j) c m | i - top < m && i - top /= 0 = False
59                       | bot - i < m && bot - i /= 0 = False
60                       | j - left < m && j - left /= 0 = False
61                       | right - j < m && right - j /= 0 = False
62                       | otherwise = True
63     where
64         (left, right, top, bot) = c ! (i,j)
65
66 spiralIndices :: Int -> Int -> [Index]
67 spiralIndices 0 _ = []
68 spiralIndices 1 w = [ (0, i) | i <- [0..(w - 1)] ]
69 spiralIndices h w = (++) (f 0 0) $ adj $ spiralIndices (h - 2) (w - 2)
70     where
71         f i j = (i,j) : l i j
72         l i j | j < w - 1 && i == 0 = f i (j + 1)
73             | j == w - 1 && i < h - 1 = f (i + 1) j
74             | j > 0 && i == h - 1 = f i (j - 1)
75             | j == 0 && i > 1 = f (i - 1) j
76             | otherwise = []
77         adj spiral = map (\(a, b) -> (a + 1, b + 1)) spiral
78
79 fillIn :: [Index] -> [Bool] -> Constraints -> Int -> [Square]
80 fillIn [] _ _ _ = []
81 fillIn _ [] _ _ = []
82 fillIn (i:is) (b:bs) c m | not b = white
83                       | checkSquare i c m = black
84                       | otherwise = white
85     where
86         white = (White '_') : fillIn is bs c m
87         black = Black : fillIn is bs (addBS i c) m
88
89 unravel :: Int -> [(Index, Square)] -> [[Square]]
90 unravel h s = map getRow [0..(h - 1)]
91     where
92         getRow i = map snd $ sortBy my_compare $ filter (isRow i) s
93         my_compare ((_,y1),_) ((_,y2),_) = compare y1 y2

```

```

94         isRow i ((x, _), _) = x == i
95
96 generateGrid :: RandomGen g => Int -> Int -> Float -> g -> Maybe [[Square]]
97 generateGrid dim m thresh seed | even dim = Just even_grid
98                               | otherwise = Nothing
99                               where
100     even_grid = half ++ rev_half
101     rev_half = reverse $ map reverse half
102     half = unravel half_h $ zip s $ squares
103     squares = fillIn s b c m
104     s = spiralIndices half_h dim
105     c = initConstraints half_h dim
106     b = map (\x -> x <= thresh) p
107     p = probs (half_h * dim) seed
108     half_h = div dim 2
109
110 printGrid :: [[Square]] -> IO ()
111 printGrid grid = do mapM_ (putStrLn) $ map show grid
112                 putStrLn ""

```
